

R	etapa4
R	etapa3
R	etapa2
R	MOTOR
S	ihmFuradeiraEstatica
R	primeiroFuro
R	segundoFuro

### Exercício 5: Carga e descarga de vagões

O desenho a seguir representa uma instalação para carga e descarga de vagões, que deverá ser automatizada.

Processo:

1. A chave S1 liga o sistema, a chave S2 desliga o sistema e H1 indica o sistema ligado;
2. Quando o vagão está posicionado o sensor S3 é atuado;
3. Quando o vagão é posicionado, ele é travado por Y2;
4. Após o vagão ser travado, a esteira (motor M1) é ligada e inicia o ciclo de enchimento do vagão;
5. A válvula YA abre o silo;
6. O pressostato P por meio do sensor S4, quando é atuado, indica que o vagão está cheio;
7. Então, fecha a válvula Yb, fechando o silo;
8. Espera 10 segundos;
9. Destrava o vagão, liberando Y2;
10. Supervisiona o sensor S3 por 20 segundos, se ele não for acionado, último vagão, desliga a esteira Motor M1.

## Fluxograma funcional




Figura 12 – Fluxograma funcional do exercício 5

## Mapa de entradas e saídas

Tabela 5 – Variáveis de entrada e saída exercício 5

Variável	Campo		
	Nome	Tipo	Endereço
Liga sistema	START	BOOL	%IX6.0
Para sistema	STOP	BOOL	%IX6.1
Detecta a presença de vagão	posicaoVagao	BOOL	%IX6.2
Detecta peso do vagão	celulaDeCarga	BOOL	%IX6.3
Mostra se motor está ligado	statusMotor	BOOL	%IX6.4
Sinal de válvula (aberto)	silAberto	BOOL	%IX6.5
Sinal de válvula (fechado)	silFechado	BOOL	%IX6.6
Lâmpada de status do sistema	lampadaStatusSistema	BOOL	%QX2.0
Sinalização do motor ligado	motorEsteira	BOOL	%QX2.1
Sinalização do silo aberto	abreSilo	BOOL	%QX2.2
Sinalização do silo fechado	fechaSilo	BOOL	%QX2.3
Liga trava vagão	travaVagao	BOOL	%QX2.3

## Tela de simulação




Figura 13 – IHM exercício 5

## Programa em Instruction List

```
PROGRAM PLC_PRG
```

```
VAR
```

```
    SISTEMA_ON: BOOL;  
    SISTEMA_OFF: BOOL;  
    tempoPosSiloFechar: TON;  
    supervisaoS3: BOOL;  
    tempoSupervisaoS3: TON;  
    inicioSupervisaoS3: BOOL;  
    aciona10seg: BOOL;  
    etapa1: BOOL;  
    etapa2: BOOL;  
    etapa3: BOOL;  
    novoVagao: R_TRIG;  
    etapa4: BOOL;
```

```
END_VAR
```

```
LD     START  
R      SISTEMA_OFF  
S      etapa1  
S      fechaSilo  
S      SISTEMA_ON
```

```
LD      STOP
R       SISTEMA_ON
S       SISTEMA_OFF
```

```
LD      SISTEMA_ON
ANDN   SISTEMA_OFF
S       supervisaoS3
S       lampadaStatusSistema
```

```
LD      etapa1
AND     posicaoVagao
AND     siloFechado
ANDN   SISTEMA_OFF
R       etapa1
S       etapa2
R       inicioSupervisaoS3
S       travaVagao
```

```
LD      etapa2
AND     travaVagao
ANDN   celulaDeCarga
ANDN   SISTEMA_OFF
S       abreSilo
R       fechaSilo
S       etapa3
R       etapa2
S       motorEsteira
```

```
LD      etapa3
AND     statusMotor
AND     siloAberto
AND     celulaDeCarga
ANDN   SISTEMA_OFF
R       etapa3
R       abreSilo
S       etapa4
S       fechaSilo
```

```
LD      etapa4
AND     siloFechado
ANDN   SISTEMA_OFF
R       etapa4
S       aciona10seg
```

```
LD      aciona10seg
ST      tempoPosSiloFechar.IN
CAL     tempoPosSiloFechar(PT := T#10000ms)
```

```
LD      tempoPosSiloFechar.Q
ANDN   SISTEMA_OFF
S       inicioSupervisaoS3
R       aciona10seg
R       travaVagao
```

LD           posicaoVagao  
ST           novoVagao.CLK  
CAL           novoVagao

LD           novoVagao.Q  
ANDN SISTEMA\_OFF  
S           etapa1

LD           inicioSupervisaoS3  
ST           tempoSupervisaoS3.IN  
CAL           tempoSupervisaoS3(PT := T#20000ms)

LD           tempoSupervisaoS3.Q  
ANDN posicaoVagao  
ANDN SISTEMA\_OFF  
R           supervisaoS3

LDN          supervisaoS3  
ANDN SISTEMA\_OFF  
R           inicioSupervisaoS3  
S           supervisaoS3  
S           etapa1  
R           motorEsteira

LD           SISTEMA\_OFF  
R           SISTEMA\_ON  
R           aciona10seg  
S           fechaSilo  
R           abreSilo  
R           lampadaStatusSistema  
R           inicioSupervisaoS3  
R           travaVagao  
R           motorEsteira